

Chapter 1

Basics

1.1 Program Structure

```
1 PROGRAM my_prog
2 !=====
3 USE module1
4 USE module2, ONLY: sub_a, sub_b, func_e
5
6 IMPLICIT NONE
7
8 CALL main()
9
10 !=====
11 CONTAINS
12 !=====
13
14 SUBROUTINE main()
15 IMPLICIT NONE
16 END SUBROUTINE main
17
18
19 SUBROUTINE other_sub()
20 IMPLICIT NONE
21 END SUBROUTINE other_sub
22
23
24 !=====
25 END PROGRAM my_prog
```

Listing 1.1: Basic program structure

1.2 Data Types

```
1 PROGRAM data_types
2 !=====
3
4 USE iso_fortran_env
5
6 IMPLICIT NONE
7
8 ! Call main subroutine
9 !#####
10 CALL main()
11
12 !=====
13 CONTAINS
14 !=====
15
16 SUBROUTINE main()
17 !#####
18 IMPLICIT NONE
19 !#####
20 REAL(kind=REAL32) :: float32 ! 32 bit signed 1/8/23 s/e/m
21 REAL(kind=REAL64) :: float64
```

```

22 INTEGER(kind=INT8) ::      int8          ! 8 bit signed
23 INTEGER(kind=INT16) ::     int16
24 INTEGER(kind=INT32) ::     int32
25 INTEGER(kind=INT64) ::     int64
26 !#####
27
28 float32 = 3.4028235d38
29 float64 = -4.29d-18
30 int8 = 127
31 int16 = 32767
32 int32 = 2147483647
33 int64 = 2**63-1
34
35 print *, "ISO FORTRAN DATA TYPES"
36 print *, float32
37 print *, float64
38 print *, int8
39 print *, int16
40 print *, int32
41 print *, int64
42
43 !#####
44 END SUBROUTINE main
45
46 !=====
47 END PROGRAM data_types

```

Listing 1.2: Using standard data types

1.3 FORTRAN - Column Major

```

1 FUNCTION sum_arrays(a, b)
2 !#####
3 REAL(kind=REAL64) ::      a(1:10, 1:4)
4 REAL(kind=REAL64) ::      b(1:10, 1:4)
5 REAL(kind=REAL64) ::      sum_arrays
6 !#####
7 INTEGER(kind=INT32) ::    i, j
8 !#####
9
10 sum_arrays = 0.0D0
11 DO j = 1, 4          ! Loop over columns (outer)
12   DO i = 1, 10        ! Loop over rows (inner)
13     sum_arrays = sum_arrays + a(i, j) + b(i, j)
14   END DO
15 END DO
16
17 ! Better solution
18 sum_arrays = 0.0D0
19 sum_arrays = SUM(a(:, :)) + SUM(b(:, :))
20
21 END FUNCTION sum_arrays

```

Listing 1.3: Column major language

Chapter 2

Output

2.1 Print

```
1 SUBROUTINE terminalprint()
2 !#####
3 IMPLICIT NONE
4 !#####
5 PRINT *, "Free text", 100, 1.4321D2
6 END SUBROUTINE terminalprint
```

Listing 2.1: Print statement

2.2 Write

```
1 SUBROUTINE terminalwrite()
2 !#####
3 IMPLICIT NONE
4 !#####
5 WRITE(*, *) "Free text"
6 WRITE(*, "(A20)") "20 character string."
7 WRITE(*, "(A24)") "20 character string."
8 WRITE(*, "(I4)") 1000
9 WRITE(*, "(I10)") 1000
10 WRITE(*, "(F20.8)") 123456789.123456789e1
11 WRITE(*, "(E20.12)") 123456789.123456789e1
12 WRITE(*, "(ES20.12)") 123456789.123456789e1
13 WRITE(*, "(EN20.12)") 123456789.123456789e1
14 WRITE(*, "(D20.12)") 123456789.123456789d1
15 WRITE(*, "(E20.12E3)") 123456789.123456789e1
16 WRITE(*, "(ES20.12E3)") 123456789.123456789e1
17 WRITE(*, "(EN20.12E3)") 123456789.123456789e1
18 END SUBROUTINE terminalwrite
```

! Integer
! Fixed Point
! Floating Point - Exponential form
! Floating Point - Scientific form
! Floating Point - Engineering form
! Double
! Floating Point - Exponential form
! Floating Point - Scientific form
! Floating Point - Engineering form

Listing 2.2: Write statement

Chapter 3

Arrays

3.1 Zero

```
1 SUBROUTINE zero(x)
2 ! #####
3 IMPLICIT NONE
4 ! #####
5 REAL(kind=REAL64), INTENT(INOUT) :: x(:, :)
6 ! #####
7 x(:, :) = 0.0d0
8 ! #####
9 END SUBROUTINE zero
```

Listing 3.1: Zero (or any number)

3.2 Linspace

```
1 ! Linspace
2
3 SUBROUTINE linspace(xmin, xmax, x)
4 ! #####
5 IMPLICIT NONE
6 ! #####
7 REAL(kind=REAL64), INTENT(IN) :: xmin
8 REAL(kind=REAL64), INTENT(IN) :: xmax
9 REAL(kind=REAL64), INTENT(INOUT) :: x(:)
10 ! #####
11 INTEGER(kind=INT32) :: n
12 REAL(kind=REAL64) :: m
13 ! #####
14 m = (xmax - xmin) / (SIZE(x,1) - 1)
15 DO n = 1, SIZE(x,1)
16   x(n) = xmin + (n - 1) * m
17 END DO
18 ! #####
19 END SUBROUTINE linspace
```

Listing 3.2: Equally spaced linear space of points, as exists in Numpy

Chapter 4

Files

4.1 Write

```
1 ! Write file
2 SUBROUTINE write ()
3 !#####
4 INTEGER(kind=INT16) ::      status, ios
5 INTEGER(kind=INT16) ::      n, m
6 INTEGER(kind=INT16) ::      lines
7 REAL(kind=REAL64) ::      arr(1:10,1:3)
8 !#####
9 ! Populate array with random floats
10 CALL RANDOM_NUMBER(arr)
11 arr = 1.0D10 * (0.5D0 - arr)
12 ! Write to file
13 OPEN(unit=99, file='out.txt')
14 DO n = 1, 10
15     WRITE(99, "(ES16.7, ES16.7, ES16.7)") arr(n, 1), arr(n, 2), arr(n, 3)
16 END DO
17 CLOSE(99)
18 END SUBROUTINE write
```

Listing 4.1: Writing to file

4.2 Read to Array

```
1 ! Read file - into array, 2 columns
2 SUBROUTINE readfile(filepath, arr)
3 !#####
4 IMPLICIT NONE
5 !#####
6 CHARACTER(LEN=*) , INTENT(IN) ::      filepath
7 REAL(kind=REAL64), ALLOCATABLE, INTENT(INOUT) ::      arr(:, :)
8 !#####
9 INTEGER(kind=INT32), PARAMETER ::      bsize = 1000000
10 CHARACTER(LEN=255) ::      line
11 CHARACTER(LEN=255) ::      buffer(1:bsize)
12 INTEGER(kind=INT32) ::      status, ios, fh
13 INTEGER(kind=INT32) ::      n
14 INTEGER(kind=INT32) ::      lines
15 LOGICAL ::      exists
16 !#####
17 ! Check file exists
18 INQUIRE (file=filepath, exist=exists)
19 IF(.NOT. exists) THEN
20     STOP "File " // TRIM(filepath) // " does not exist"
21 END IF
22 ! Count Lines & Read To Buffer
23 lines = 0
24 OPEN(newunit=fh, file=filepath, action='read', iostat=status)
25 DO n = 1, bsize
26     READ(fh, "(A255)", IOSTAT=ios) line
27     IF(TRIM(line) .NE. "") THEN
28         lines = lines + 1
```

```

29     buffer(lines) = line
30 END IF
31 If(ios /= 0)Then
32     EXIT
33 End If
34 END DO
35 CLOSE(fh)
36 ! Allocate array
37 IF(ALLOCATED(arr)) DEALLOCATE(arr)
38 ALLOCATE(arr(1:lines, 1:2))
39 ! Read into array
40 DO n = 1, lines
41     READ(buffer(n), "(10F8.2,10F8.2)", IOSTAT=ios) arr(n, 1), arr(n, 2)
42 END DO
43 END SUBROUTINE readfile

```

Listing 4.2: Read file into a 2D, 2 column, array

4.3 Read Namelist

```

1  ! computer.nml
2  !
3  ! &COMPUTER
4  ! os='Ubuntu'
5  ! ram=64
6  ! proc%make='AMD'
7  ! proc%model='Ryzen 9'
8  ! proc%freq=3.8
9  ! proc%cores=12
10 ! proc%cache(1)=0.75
11 ! proc%cache(2)=6
12 ! proc%cache(3)=64
13 ! drive(1)=256
14 ! drive(2)=5000
15 ! drive(3)=5000
16 ! drive(4)=0
17 ! drive(5)=0
18 ! /
19
20 SUBROUTINE readnamelist(filepath)
21 !#####
22 IMPLICIT NONE
23 !#####
24 CHARACTER(LEN=*), INTENT(IN) ::          filepath
25 !#####
26 TYPE :: t_processor
27     CHARACTER(len=16) ::          make
28     CHARACTER(len=16) ::          model
29     INTEGER(kind=INT32) ::        cores
30     REAL(kind=REAL32) ::          freq
31     REAL(kind=REAL32) ::          cache(1:3)
32 END TYPE t_processor
33 CHARACTER(len=16) ::          os
34 INTEGER(kind=INT32) ::        ram
35 TYPE(t_processor) ::          processor
36 INTEGER(kind=INT32) ::        drive(1:5)
37 LOGICAL ::          exists
38 INTEGER(kind=INT32) ::        status
39 INTEGER(kind=INT32) ::        fh
40 !#####
41 NAMELIST /COMPUTER/ os, ram, processor, drive ! Define namelist
42 !#####
43 ! Check file exists
44 INQUIRE (file=filepath, exist=exists)
45 IF(.NOT. exists)THEN
46     STOP "File " // TRIM(filepath) // " does not exist"
47 END IF
48 ! Open and read file
49 OPEN (action='read', file=filepath, iostat=status, newunit=fh)
50 read (nml=COMPUTER, iostat=status, unit=fh)
51 CLOSE(fh)
52 ! Check read successful
53 IF(status .NE. 0)THEN
54     STOP "File " // TRIM(filepath) // " contains an incorrect namelist format"

```



```
55 END IF
56 ! Output
57 WRITE (*, "(A20)") ADJUSTR(os)
58 WRITE (*, "(I20)") ram
59 WRITE (*, "(A20)") ADJUSTR(processor%make)
60 WRITE (*, "(A20)") ADJUSTR(processor%model)
61 WRITE (*, "(I20)") processor%cores
62 WRITE (*, "(I20)") drive(1)
63 WRITE (*, "(I20)") drive(2)
64 WRITE (*, "(I20)") drive(3)
65 WRITE (*, "(I20)") drive(4)
66 WRITE (*, "(I20)") drive(5)
67 END SUBROUTINE readnamelist
```

Listing 4.3: Read namelist

Chapter 5

Interpolation

5.1 Lagrange

```
1  ! Lagrange interpolation
2  !
3  FUNCTION interpn(xi, x, y)
4  !#####
5  IMPLICIT NONE
6  !#####
7  REAL(kind=REAL64), INTENT(IN) :: xi
8  REAL(kind=REAL64), INTENT(IN) :: x(:)
9  REAL(kind=REAL64), INTENT(IN) :: y(:)
10 REAL(kind=REAL64) :: interpn
11 !#####
12 REAL(kind=REAL64) :: li
13 INTEGER(kind=INT32) :: xsize
14 INTEGER(kind=INT32) :: i, j
15 !#####
16 IF(SIZE(x) .NE. SIZE(y)) THEN
17   STOP "Error interpn() unequal array sizes"
18 END IF
19 ! Set values
20 interpn = 0.0D0
21 xsize = SIZE(x)
22 ! Loop
23 DO i = 1, xsize
24   li = 1.0D0
25   DO j = 1, xsize
26     IF(i .NE. j) THEN
27       li = li * (xi - x(j)) / (x(i) - x(j))
28     END IF
29   END DO
30   interpn = interpn + li * y(i)
31 END DO
32 END FUNCTION interpn
```

Listing 5.1: Interpolate small data set

5.2 Lagrange - Larger Data Set

```
1  SUBROUTINE interpfill(x_in, y_in, k, x_out, y_out)
2  !#####
3  IMPLICIT NONE
4  !#####
5  REAL(kind=REAL64), INTENT(IN) :: x_in(:)
6  REAL(kind=REAL64), INTENT(IN) :: y_in(:)
7  INTEGER(kind=INT32), INTENT(IN) :: k
8  REAL(kind=REAL64), INTENT(INOUT) :: x_out(:)
9  REAL(kind=REAL64), INTENT(INOUT) :: y_out(:)
10 !#####
11 INTEGER(kind=INT32) :: s_in, s_out
12 INTEGER(kind=INT32) :: n, nn, a, b, i, j
13 REAL(kind=REAL64) :: m
14 REAL(kind=REAL64) :: li
```

```

15 !#####
16 IF(SIZE(x_in) .NE. SIZE(y_in) .OR. SIZE(x_out) .NE. SIZE(y_out))THEN
17   STOP "Error interpolate() unequal array sizes"
18 END IF
19 ! Array sizes
20 s_in = SIZE(x_in)
21 s_out = SIZE(x_out)
22 ! Fill X
23 m = (x_in(s_in) - x_in(1)) / (s_out - 1)
24 DO n = 1, s_out
25   x_out(n) = x_in(1) + (n - 1) * m
26 END DO
27 ! Fill Y
28 y_out(:) = 0.0d0
29 nn = 1
30 DO n = 1, s_out
31   DO WHILE(nn .LT. s_in .AND. .NOT. (x_out(n) .GE. x_in(nn) .AND. x_out(n) .LE. x_in(nn+1)))
32     nn = nn + 1
33   END DO
34   ! Choose subset of points for interpolation
35   a = MIN(nn, s_in - k + 1)
36   b = a + k - 1
37   DO i = a, b
38     li = 1.0d0
39     DO j = a, b
40       IF(i .NE. j) THEN
41         li = li * (x_out(n) - x_in(j)) / (x_in(i) - x_in(j))
42       END IF
43     END DO
44     y_out(n) = y_out(n) + li * y_in(i)
45   END DO
46 END DO
47 END SUBROUTINE interpfill

```

Listing 5.2: Interpolate with k points over a larger data set

5.3 Lagrange - Gradient

```

1 !
2 ! Lagrange interpolation
3 !
4 FUNCTION interpnggrad(xi, x, y)
5 !#####
6 IMPLICIT NONE
7 !#####
8 REAL(kind=REAL64), INTENT(IN) :: xi
9 REAL(kind=REAL64), INTENT(IN) :: x(:)
10 REAL(kind=REAL64), INTENT(IN) :: y(:)
11 REAL(kind=REAL64) :: interpnggrad
12 !#####
13 REAL(kind=REAL64) :: fx, gx, psum
14 INTEGER(kind=INT32) :: xsize
15 INTEGER(kind=INT32) :: i, j, k
16 !#####
17 IF(SIZE(x) .NE. SIZE(y))THEN
18   STOP "Error interpnggrad() unequal array sizes"
19 END IF
20 ! Set values
21 xsize = SIZE(x,1)
22 interpnggrad = 0.0d0
23 DO i=1,SIZE(x,1)
24   fx = 1.0d0
25   gx = 0.0d0
26   DO j=1,SIZE(x,1)
27     IF(i .NE. j) THEN
28       fx = fx / (x(i) - x(j))
29       psum = 1.0d0
30       DO k=1,SIZE(x,1)
31         IF((i .NE. k) .AND. (j .NE. k))THEN
32           psum = psum * (xi - x(k))
33         END IF
34       END DO
35       gx = gx + psum
36     END IF

```

```
37  END DO
38  interpnggrad = interpnggrad + fx * gx * y(i)
39 END DO
40 END FUNCTION interpnggrad
```

Listing 5.3: Gradient at xi using a small data set

Chapter 6

Differentiation

6.1 Zero

```
1 FUNCTION grad(f_in, x)
2 !#####
3 IMPLICIT NONE
4 !#####
5 REAL(kind=REAL64), EXTERNAL ::      f_in
6 REAL(kind=REAL64), INTENT(IN) ::    x
7 REAL(kind=REAL64) ::                grad
8 !#####
9 REAL(kind=REAL64) ::                h
10 !#####
11 h = 1.0D-5
12 grad = (f_in(x+h) - f_in(x-h)) / (2 * h)
13 END FUNCTION grad
```

Listing 6.1: Calculate 1st derivative

6.2 Linspace

```
1 FUNCTION grad2(f_in, x)
2 !#####
3 IMPLICIT NONE
4 !#####
5 REAL(kind=REAL64), EXTERNAL ::      f_in
6 REAL(kind=REAL64), INTENT(IN) ::    x
7 REAL(kind=REAL64) ::                grad2
8 !#####
9 REAL(kind=REAL64) ::                h
10 !#####
11 h = 1.0D-5
12 grad2 = (f_in(x+h) - 2.0d0 * f_in(x) + f_in(x-h)) / h**2
13 END FUNCTION grad2
```

Listing 6.2: Calculate 2nd derivative

Chapter 7

Integration

7.1 Simpson Integration Part 1

```
1  ! Simpson integration
2  ! Must have an even number of intervals
3  ! (odd number of data points)
4  !
5  FUNCTION integrate(x, y)
6  !#####
7  IMPLICIT NONE
8  !#####
9  REAL(kind=REAL64), INTENT(IN) :: x(:)
10 REAL(kind=REAL64), INTENT(IN) :: y(:)
11 REAL(kind=REAL64) :: integrate
12 !#####
13 REAL(kind=REAL64) :: h, odd, even
14 INTEGER(kind=INT32) :: xsize
15 INTEGER(kind=INT32) :: n
16 !#####
17 IF(SIZE(x) .NE. SIZE(y))THEN
18   STOP "Error integrate() unequal array sizes"
19 END IF
20 odd = 0.0d0
21 even = 0.0d0
22 xsize = SIZE(x, 1)
23 h = (x(xsize) - x(1)) / (xsize - 1)
24 ! Odd and even terms
25 n = 1
26 DO WHILE(n < xsize)
27   n = n + 1
28   even = even + y(n)
29   n = n + 1
30   odd = odd + y(n)
31 END DO
32 ! Finish
33 integrate = (h / 3.0d0) * (y(1) + y(xsize) + 4.0d0 * even + 2.0d0 * odd)
34 END FUNCTION integrate
```

Listing 7.1: Using Simpson Integration

7.2 Simpson Integration Part 2

```
1  ! Simpson integration
2  ! Interpolates to odd number of points
3  ! Requires interpfill subroutine
4  !
5  FUNCTION integrate(x, y)
6  !#####
7  IMPLICIT NONE
8  !#####
9  REAL(kind=REAL64), INTENT(IN) :: x(:)
10 REAL(kind=REAL64), INTENT(IN) :: y(:)
11 REAL(kind=REAL64) :: integrate
12 !#####
```

```

13 REAL(kind=REAL64) ::          h, odd, even
14 INTEGER(kind=INT32) ::        xsize
15 INTEGER(kind=INT32) ::        n
16 REAL(kind=REAL64), ALLOCATABLE :: xt(:)
17 REAL(kind=REAL64), ALLOCATABLE :: yt(:)
18 !#####
19 odd = 0.0d0
20 even = 0.0d0
21 IF(SIZE(x) .NE. SIZE(y))THEN
22   STOP "Error integrate() unequal array sizes"
23 END IF
24 IF(MOD(SIZE(x, 1), 2) .EQ. 0)THEN
25   ALLOCATE(xt(1:SIZE(x, 1)+1))
26   ALLOCATE(yt(1:SIZE(y, 1)+1))
27   CALL interpfill(x, y, 3, xt(:), yt(:))
28   xsize = SIZE(xt, 1)
29   h = (xt(xsize) - xt(1)) / (xsize - 1)
30   ! Odd and even terms
31   n = 1
32   DO WHILE(n < xsize)
33     n = n + 1
34     even = even + yt(n)
35     n = n + 1
36     odd = odd + yt(n)
37   END DO
38 ELSE
39   xsize = SIZE(x, 1)
40   h = (x(xsize) - x(1)) / (xsize - 1)
41   ! Odd and even terms
42   n = 1
43   DO WHILE(n < xsize)
44     n = n + 1
45     even = even + y(n)
46     n = n + 1
47     odd = odd + y(n)
48   END DO
49 END IF
50 ! Finish
51 integrate = (h / 3.0d0) * (y(1) + y(xsize) + 4.0d0 * even + 2.0d0 * odd)
52 END FUNCTION integrate

```

Listing 7.2: Using Simpson Integration

Chapter 8

OpenMP

8.1 Summing in parallel

```
1  !
2  ! gfortran -fopenmp -o sum.x sum.f90 && ./sum.x
3  !
4
5  ! Possibly not efficient, over head may outweigh benefit
6
7  SUBROUTINE parallelsun()
8  !#####
9  IMPLICIT NONE
10 !#####
11 REAL(kind=REAL64) :: myarray(1:100000)
12 REAL(kind=REAL64) :: s_thread, s_total
13 INTEGER(kind=INT64) :: n, tid, tcount
14 INTEGER(kind=INT64) :: OMP_GET_NUM_THREADS, OMP_GET_THREAD_NUM
15 !#####
16 s_total = 0.0d0
17 myarray(1:100000) = 1.0D0
18 !$OMP PARALLEL &
19 !$OMP PRIVATE(n, s_thread, tid) &
20 !$OMP SHARED(myarray, s_total)
21 s_thread = 0.0d0
22 !$OMP DO
23 DO n = 1, 100000
24   s_thread = s_thread + myarray(n)
25 END DO
26 !$OMP END DO
27 !$OMP CRITICAL
28 tcount = OMP_GET_NUM_THREADS()
29 tid = OMP_GET_THREAD_NUM()
30 s_total = s_total + s_thread
31 WRITE(*,*) tid, "/", tcount, " ", s_thread
32 !$OMP END CRITICAL
33 !$OMP END PARALLEL
34 WRITE(*,*) s_total
35 END SUBROUTINE parallelsun
```

Listing 8.1: Using OpenMP to sum over several threads